```
Maxima 5.13.0 http://maxima.sourceforge.net
Using Lisp GNU Common Lisp (GCL) GCL 2.6.7 (aka GCL)
Distributed under the GNU Public License. See the file COPYING.
Dedicated to the memory of William Schelter.
This is a development version of Maxima. The function bug_report()
provides bug reporting information.
```

This is a example of redoing a Maple worksheet using free software Maxima.

```
(%i1) an:x^n;
      bn:1/n^2;
```

(%o124) x^n

(%o125) $\frac{1}{n^2}$

```
(%i126) A1:sum(an,n,1,inf);
        B1:sum(bn,n,1,inf);
```

(%o126) $\sum_{n=1}^{\infty} x^n$

(%o127) $\sum_{n=1}^{\infty} \frac{1}{n^2}$

```
(%i128) A2:A1, simpsum=true;
```

Is $|x|-1$ positive, negative, or zero?`negative`

(%o128) $\frac{x}{1-x}$

```
(%i129) B2:B1, simpsum=true;
```

(%o129) $\frac{\pi^2}{6}$

```
(%i130) subst(x=1/3,A1=A2);
```

(%o130) $\sum_{n=1}^{\infty} \frac{1}{3^n} = \frac{1}{2}$

```
(%i131) S1:sum(an*bn,n,1,inf);
```

(%o131) $\sum_{n=1}^{\infty} \frac{x^n}{n^2}$

```
(%i132) S2:S1, simpsum=true;
```

(%o136) $\sum_{n=1}^{\infty} \frac{x^n}{n^2}$

Remove the n^2 in the denominator by repeatedly differentiating.

```
(%i137) T1:diff(S1,x);
```

(%o137) $\sum_{n=1}^{\infty} \frac{x^{n-1}}{n}$

```
(%i138) T2:intosum(x*T1);
```

(%o138) $$\sum_{n=1}^{\infty} \frac{x^n}{n}$$

```
(%i139) T3:diff(T2,x);
```

(%o139) $$\sum_{n=1}^{\infty} x^{n-1}$$

```
(%i140) T4:T3, simpsum=true;
```

Is $|x|-1$ positive, negative, or zero?`negative`

(%o140) $$\frac{1}{1-x}$$

Therefore $(xy')' = 1/(1-x)$ where $y = \sum_{n=1}^{\infty} x^n n^{-2}$. We now integrate to obtain y.

```
(%i141) T5:integrate(T4,x,0,t);
```

Is t positive, negative, or zero?`negative`

Is $x+t-1$ positive or negative?`negative`

(%o141) $$-\log(1-t)$$

```
(%i142) T6:T5/t;
```

(%o142) $$-\frac{\log(1-t)}{t}$$

```
(%i143) T7:integrate(T6,t,0,x);
```

Is x positive, negative, or zero?`negative`

(%o143) $$-\log(1-x)\log(x) - \mathrm{li}_2(1-x) + \frac{\pi^2}{6}$$

```
(%i144) T8:subst(x=1/3,T7);
```

(%o144) $$\log\left(\frac{2}{3}\right)\log(3) - \mathrm{li}_2(\frac{2}{3}) + \frac{\pi^2}{6}$$

```
(%i145) float(T8);
```

(%o145) 0.36621322997706

Sum the first 10 terms to make an approximation. Note that this approximation agrees to the value computed above to the first 5 decimals.

```
(%i146) S3:sum(an*bn,n,1,10);
```

(%o146) $$\frac{x^{10}}{100}+\frac{x^9}{81}+\frac{x^8}{64}+\frac{x^7}{49}+\frac{x^6}{36}+\frac{x^5}{25}+\frac{x^4}{16}+\frac{x^3}{9}+\frac{x^2}{4}+x$$

```
(%i147) S4:subst(x=1/3,S3);
```

(%o147) $$\frac{45774786439}{124994923200}$$

```
(%i148) float(S4);
```

(%o148) 0.36621316503997

(%i149)